Función que define la estrategia 

El éxito de cualquier estrategia depende principalmente de los criterios que se tengan para operar. La función que define estos criterios en la mayoría de los casos es la más importante e imprescindible de todas. Según la estrategia, la función puede devolver valores que correspondan con unos criterios comerciales particulares. 
En la mayoría de los casos, se pueden definir los siguientes criterios:

· Criterio para abrir una orden;

· Criterio para cerrar una orden;

· Criterio para cerrar una parte de una orden;

· Criterio para cerrar ordenes opuestas;
· Criterio para modificar el StopLoss de una orden;
· Criterio para colocar una orden pendiente;

· Criterio para cancelar una orden pendiente;

· Criterio para modificar el precio de apertura de una orden pendiente;

· Criterio para modificar el StopLoss de una orden pendiente;

En la mayoría de los casos, la activación de un criterio comercial es exclusiva en relación a otros criterios comerciales. In most cases, the triggering of one trading criterion is exclusive as related to other trading criteria. Por ejemplo, si el criterio para abrir una orden de compra se vuelve importante en un momento dado, esto significará que los criterios usados para cerrar órdenes de compra o abrir órdenes de venta, no pueden ser importantes en ese mismo momento (véase la relación de criterios comerciales LINK). Al mismo tiempo, según las normas inherentes de una estrategia dada, algunos criterios se pueden activar simultáneamente. Por ejemplo, los criterios para cerrar una orden de venta y los criterios para modificar una orden pendiente BuyStop, se pueden dar al mismo tiempo. 

Una estrategia comercial impone requisitos al contenido y a la tecnología que usa la función que define los criterios comerciales. Una función solo puede devolver un valor. Por lo tanto, si la estrategia comercial de un asesor experto implica utilizar criterios comerciales mutuamente excluyentes, el valor que devuelve la función puede estar asociado a uno de esos criterios. Sin embargo, si la estrategia permite activar varios criterios al mismo tiempo, sus valores deben ser enviados por medio de otras funciones para que sean procesados, utilizando las variables globales para ello. 

La estrategia comercial que veremos en el EA más abajo, solo permite la utilización de criterios mutuamente excluyentes. Por esta razón la función que utilizaremos para definir los criterios, llamada Criterion(), se comunicara con otras funciones, por medio del valor que esta devuelva . 

Función definida por el usuario Criterion()
int Criterion()
Esta función calcula los criterios comerciales. Puede devolver los siguientes valores:

10 - accionó un criterio comercial para abrir una orden de compra;

20 - accionó un criterio comercial para abrir una orden de venta;

11 - accionó un criterio comercial para cerrar una orden de compra;

21 - accionó un criterio comercial para cerrar una orden de venta;

0 – No hay criterios importantes disponibles;

-1 – la divisa usada no es EURUSD.

La función utiliza las siguientes variables externas:
St_min - el nivel inferior del oscilador estocástico;

St_max - el nivel superior del oscilador estocástico;

Open_Level - el nivel del indicador MACD (para abrir una orden);

Close_Level - el nivel del indicador MACD (para cerrar una orden).

Para mostrar los mensajes, la función utilizara la función Inform(). Si esta no está incluida los mensajes no se verán en pantalla. 

El siguiente es el código de la función Criterion() que definirá los criterios comerciales, la que guardaremos en un archivo llamado Criterion.mqh y que luego incluiremos en nuestro proyecto. 

//-------------------------------------------------------------------------

// Criterion.mqh

// Este código solo debe ser usado de manera educacional.

//-------------------------------------------------------------------- 1 --

// Función que calcula los criterios comerciales.

// Valores devueltos:

// 10 – Abriendo compra 

// 20 – Abriendo venta

// 11 – Cerrando compra

// 21 – Cerrando venta

// 0  - no hay criterio importante disponible

// -1 – se esta utilizando otra divisa

//-------------------------------------------------------------------- 2 --

// Variables externas:

extern int St_min=30;                  // Nivel mínimo de estocástico

extern int St_max=70;                  // Nivel máximo de estocástico

extern double Open_Level =5;           // MACD nivel para abrir orden (+/-)

extern double Close_Level=4;           // MACD nivel para cerrar orden (+/-)

//-------------------------------------------------------------------- 3 --

int Criterion()                        // Función definida por el usuario
  {

   string Sym="EURUSD";

   if (Sym!=Symbol())                  // Si la divisa es diferente a EURUSD
     {

      Inform(16);                      // Mensaje..

      return(-1);                      // .. y salida
     }

   double

   M_0, M_1,                           // Valor PRINCIPAL en vela 0 y 1

   S_0, S_1,                           // Valor SEÑAL en vela 0 y 1

   St_M_0, St_M_1,                     // Valor PRINCIPAL en vela 0 y 1
   St_S_0, St_S_1;                     // Valor SEÑAL en vela 0 y 1
   double Opn=Open_Level*Point;        // Apertura de nivel de MACD (pips)

   double Cls=Close_Level*Point;       // Cierre de nivel de MACD (pips)

//-------------------------------------------------------------------- 4 --

   // Parámetros de los indicadores técnicos:

   M_0=iMACD(Sym,PERIOD_H1,12,26,9,PRICE_CLOSE,MODE_MAIN,0); // 0 vela
   M_1=iMACD(Sym,PERIOD_H1,12,26,9,PRICE_CLOSE,MODE_MAIN,1); // 1 vela
   S_0=iMACD(Sym,PERIOD_H1,12,26,9,PRICE_CLOSE,MODE_SIGNAL,0);//0 vela
   S_1=iMACD(Sym,PERIOD_H1,12,26,9,PRICE_CLOSE,MODE_SIGNAL,1);//1 vela
   St_M_0=iStochastic(Sym,PERIOD_M15,5,3,3,MODE_SMA,0,MODE_MAIN,  0);

   St_M_1=iStochastic(Sym,PERIOD_M15,5,3,3,MODE_SMA,0,MODE_MAIN,  1);

   St_S_0=iStochastic(Sym,PERIOD_M15,5,3,3,MODE_SMA,0,MODE_SIGNAL,0);

   St_S_1=iStochastic(Sym,PERIOD_M15,5,3,3,MODE_SMA,0,MODE_SIGNAL,1);

//-------------------------------------------------------------------- 5 --

   // Calculando el criterio de operación
   if(M_0>S_0 && -M_0>Opn && St_M_0>St_S_0 && St_S_0<St_min)

      return(10);                      // Abriendo compra 

   if(M_0<S_0 &&  M_0>Opn && St_M_0<St_S_0 && St_S_0>St_max)

      return(20);                      // Abriendo venta
   if(M_0<S_0 &&  M_0>Cls && St_M_0<St_S_0 && St_S_0>St_max)

      return(11);                      // Cerrando compra  

   if(M_0>S_0 && -M_0>Cls && St_M_0>St_S_0 && St_S_0>St_min)

      return(21);                      // Cerrando venta 

//-------------------------------------------------------------------- 6 --

   return(0);                          // Salimos de la función definida por el usuario
  }

//-------------------------------------------------------------------- 7 –
En el bloque 1 – 2, se explica los valores devueltos por la función. En el bloque 2 – 3, se declaran algunas variables externas. El archivo incluido Criterion.mqh es el único archivo usado en este EA, en el cual (en este caso externa) se declaran variables globales. En la sección llamada estructura de un programa normal, se pueden encontrar razones para declarar todas las variables globales sin excepción en un archivo separado Variables.mqh. En este caso, las variables externas se declaran en el archivo Criterion.mqh por dos razones: primero, para demostrar que es técnicamente posible (no siempre es posible); y en segundo lugar para mostrar cómo utilizar variables externas en la depuración y las pruebas de un programa. 
Es técnicamente posible declarar variables externas en el archivo Criterion.mqh, porque estas variables no se utilizarán en otras funciones del programa. Los valores de las variables externas declaradas en el bloque 2-3 determinan los niveles para el oscilador estocástico y el indicador MACD y se utilizaran solamente en la función Criterion(). La declaración de variables externas en el archivo que contiene la función que define los criterios comerciales puede ser razonable, si el archivo se utiliza temporalmente para la depuración y cálculo de los valores de esas variables en el programa. Con este fin, se puede agregar otras variables externas en el programa, por ejemplo, para optimizar las entradas del indicador (en este caso, las constantes fijadas en los valores 12,26,9 para MACD y 5,3,3 para el oscilador estocástico).  Una vez hechas las optimizaciones de las variables, usted puede suprimir estas variables externas del programa y sustituirlas por constantes con los valores calculados en la optimización. 
En el bloque 3-4, se abren y describen las variables locales. El EA está pensado para que sea utilizado en la divisa “EURUSD”, así que en el bloque 3 – 4 se revisa si esta condición se cumple. Si el EA se pone en marcha en otra divisa, la función termina su ejecución y devuelve el valor -1 (divisa incorrecta).  

En el programa, los valores de dos indicadores se calculan en la vela actual y en la anterior (bloque 4-5). Generalmente, cuando se utiliza el oscilador estocástico y MACD, las señales para comprar o vender se dan cuando dos líneas del indicador se encuentran entre sí.  En este caso, utilizamos dos indicadores simultáneamente para definir los criterios comerciales. La probabilidad de que se crucen simultáneamente líneas de dos indicadores es algo baja. Es mucho más probable que se crucen una por una – primero en un indicador, y un rato más adelante en otro. Si las líneas de estos dos indicadores se cruzan dentro de un corto periodo de tiempo,  se considera que se ha formado un criterio comercial. 
Por ejemplo, a continuación se muestra cómo se calcula un criterio comercial para hacer una compra (bloque 5-6):
if(M_0 > S_0 && - M_0 > Opn && St_M_0>St_S_0 && St_S_0<St_min)
Según el código anterior, el criterio para abrir una orden de compra se forma cuando se cumplen las siguientes condiciones. 

· En el indicador MACD, la línea PRINCIPAL (histograma) del indicador está por encima de la línea SEÑAL del indicador y debajo del nivel más bajo Open_Level (Fig. 157);
· En el Oscilador Estocástico, la línea PRINCIPAL  (histograma) está por encima de la línea SEÑAL del indicador y debajo del nivel más bajo St_min (Fig.158). 

Fig. 157. Condiciones necesarias de las líneas del indicador MACD para confirmar la importancia de los criterios para abrir y cerrar órdenes. 
En la parte izquierda de la fig. 157, se muestran las posiciones de las líneas del indicador MACD, en las cuales dos criterios se accionan – abrir orden de compra y cerrar orden de venta. La línea PRINCIPAL del indicador esta debajo del nivel 0.0005 en el transcurso de T1 = t 1 - t 0. Si el oscilador estocástico también da una señal de entrada en este momento, se activara el criterio para abrir una orden de compra. En él transcurso de T2 = t 2 - t 0, la línea PRINCIPAL está debajo del nivel 0.0004.  Si el oscilador estocástico también da una señal de salida en este momento, se activara el criterio para cerrar una orden de venta. 
Observe por favor que, en T1 se activan dos criterios (si son confirmados por el oscilador estocástico). Antes habíamos mencionado que la función Criterion() devuelve un solo valor, a saber, el valor que se asigna al criterio activado. Durante este periodo, hay que elegir uno de los dos criterios. Este problema se debe solucionar por adelantado en el programa, según las reglas de la estrategia comercial. 

En este caso (según la estrategia comercial considerada), la prioridad de abrir una orden de compra es mayor que la de cerrar una orden de venta.  Es por esta razón que en el bloque 5 – 6, la línea de código que contiene la orden de compra se coloca por encima de las demás líneas de código. Si durante el período de T1 (fig. 157), tenemos la confirmación del oscilador estocástico, la función devolvería 10, que representa el criterio de compra. Luego dentro del periodo T1 a t2, la función devolverá 21, asignado al criterio para cerrar una venta. 

Al  mismo tiempo, en la función que maneja las peticiones comerciales, se formará una petición comercial. En primer lugar se accionara el criterio para abrir una orden de compra, en segundo, se accionara el criterio para cerrar todas las órdenes de venta. Tan pronto como no haya ordenes, se solicitara una orden de compra. Respectivamente, cuando el criterio para cerrar las órdenes de venta se accione, se ejecutara una secuencia que cerrará todas las órdenes de venta abiertas. (ver  función de comercio LINK).
Las condiciones en que el Oscilador Estocástico da confirmación se muestra en la figura 158.

Fig. 158. Condiciones necesarias de las líneas del Oscilador Estocástico para confirmar la importancia de los criterios para abrir y cerrar órdenes. 

Según el código de programa escrito en el bloque 5-6, los criterios para abrir una compra y cerrar una venta se activaran, siempre que la línea PRINCIPAL del indicador, está por encima de la línea SEÑAL en el Oscilador Estocástico, la línea PRINCIPAL está por debajo del nivel mínimo St_min. En la fig. 158, estas condiciones se dan en el periodo TS. Los criterios inversos se dan en el segundo Ts, donde se forman los criterios comerciales para abrir una orden de venta y cerrar una orden de compra. 

Es importante señalar que en esta estrategia, el indicador MACD esta funcionando en graficas de 1 hora, mientras que el oscilador estocástico funciona en graficas de 15 minutos. El tiempo de las graficas se puede cambiar mientras se está optimizando y perfeccionando la estrategia. Sin embargo, después de hacer estas pruebas, y tener los valores optimizados en la función Criterion(), es necesario convertir estos valores en constantes para todos los parámetros calculados incluyendo el tiempo de las graficas. El EA debe ser utilizado bajo las condiciones en las que se ha creado. En el ejemplo anterior (con los valores de los periodos PERIOD_H1 y PERIOD_M15 especificados explícitamente en los indicadores), el EA tomara las mismas decisiones independientemente del periodo del grafico actual de la divisa donde está funcionando el EA. 
Los criterios comerciales utilizados en este EA están hechos para propósitos de enseñanza y no deben ser utilizados en cuentas reales. 

